

TOMA DE DECISIONES CON GAMS¹

1.-Introducción al programa GAMS

El programa GAMS (General Algebraic Modeling System) es un software desarrollado por A. Brooke, D. Kendrick y A. Meeraus. A diferencia de otros paquetes de software de implementación de algoritmos matemáticos que permiten resolver los problemas de optimización, el programa GAMS presenta la ventaja de plantear un lenguaje de modelización que permite poder escribir en un editor la formulación matemática del problema y posteriormente aplicarle una serie de “solvers” o programas de resolución. Además permite también trabajar con el mismo lenguaje en otro tipo de plataformas no basadas en Windows como pueden ser UNIX, LINUX, etc. También permite la interrelación con otros lenguajes de programación.

En primer lugar para poder trabajar con el programa GAMS, lo primero que necesitamos es disponer de una versión de dicho programa. Se puede obtener una versión *demo* del programa conectando a la página web

<http://www.gams.com/>

Las limitaciones de la versión gratuita son las siguientes

Máximo numero de filas:	300
Máximo numero de columnas:	300
Máximo numero de elementos distintos de cero:	2000
Máximo numero de elementos no lineales:	1000
Máximo numero de variables discretas:	50


Esta versión restringida será suficiente para el objetivo que pretendemos pues como vemos nos permitirá trabajar con hasta 300 variables y 300 restricciones. Si se desea resolver problemas de mayor tamaño, es necesario adquirir una licencia de una versión no restringida.

¹ Este material y otro material complementario se puede encontrar en <http://www.uv.es/mmocho/CGyF>

2.- Instalación del programa.

Una vez bajado el fichero de la web de GAMS, la instalación del programa es muy sencilla, pues basta con ejecutar dicho fichero y comenzará la instalación automática del programa, únicamente habrá que elegir el directorio en el cual queremos instalarlo, por defecto *c:\archivos de programa\gamsxx.x* (donde xx.x es la versión correspondiente)

Finalizada la instalación se accede a dicho programa desde el icono creado durante la instalación en el escritorio o bien desde *inicio, programas*, apareciendo entonces la siguiente ventana




Para empezar es conveniente crear un proyecto mediante el menú *File, Project, new Project* teclear un nombre y guardarlo en la carpeta que queramos. Por defecto el proyecto se crea en la instalación en

c:\users\xxxxxx\documents\gamsdir\projdir\gmsproj.gpr

El proyecto, sirve para indicarle al programa la carpeta de trabajo y la configuración y en consecuencia donde deberá enviar todos los ficheros de ejecución y resultados.

A continuación ya podremos crear nuestro fichero, mediante el editor del gams, lo cual se realiza eligiendo de la barra de menús la opción *file* y de ésta *new*, realizado esto aparecerá la siguiente pantalla



Como se puede apreciar, se genera un fichero denominado `untitled_1.gms`, por lo que resulta conveniente antes de empezar a teclear nuestro modelo asignarle nombre a nuestro fichero, mediante la opción *file, save as*,

3.- Comandos de GAMS.

Para poder introducir un modelo, previamente necesitamos conocer algunas nociones de los comandos básicos de gams. Dentro de los bloques que podemos utilizar, hay que distinguir entre los obligatorios que necesariamente habrá que introducir para que el programa funcione y los optativos que, aunque no son necesarios para la ejecución del modelo, nos pueden facilitar el trabajo y la lectura del fichero.

3.1 Bloques optativos

a) Inclusión de texto

En cualquier parte del fichero podemos introducir notas aclaratorias sobre el mismo o cualquier comentario, que no será compilado, de dos formas distintas.

1) Escribiendo un asterisco `*` en la primera columna de una línea, dicha línea será de comentario y no será compilada. Con este comando se precisa poner `*` en cada línea de comentario, por lo que sólo se utiliza para comentarios muy cortos.

** Ejemplo de texto con asterisco*

2) Con los comandos \$ontext y \$offtext de la siguiente forma.

\$ontext

Ejemplo de selección de proyectos de inversión repetitivos y no repetitivos con limitaciones presupuestarias

\$offtext

Se utiliza para comentarios de más de una línea o incluso para habilitar o deshabilitar parte de un fichero que no queremos utilizar en un determinado momento.

Todo lo que se encuentre entre ambos comando será considerado como texto y no se utilizará para la ejecución

b) Definición de conjuntos

La definición de conjuntos es una herramienta muy útil para la construcción de modelos en GAMS, puesto que permite compaginar el uso de nombres completos para referirnos a las distintas opciones con la brevedad del lenguaje matemático.

La forma de definir los Conjuntos es la siguiente

```
SET I Inversiones /Proy1,Proy2,Inv1,Inv2,Maq,Edif./  
P proyectos /P1*P15/  
A años /A1*A9/;
```

Se introducen con el comando **SET**, a continuación el nombre del conjunto (índice), un comentario (opcional) si lo creemos oportuno y entre / los nombres de los elementos separados por comas. La segunda forma es una forma abreviada que se puede emplear cuando los elementos siguen una serie, podríamos haber utilizado /a1*a10/. La definición de conjuntos debe terminar con ;

También se pueden definir subconjuntos de otros conjuntos definidos anteriormente

```
SET PB(P) proyectos no repetitivos /P1*P8/  
PC(P) proyectos repetitivos /P9*P15/;
```

c) Introducción de datos

Existen tres tipos de comandos para introducir datos:

SCALAR se utilizan para declarar o inicializar una constante y se introducen con el comando scalar seguido del nombre que se quiere dar, un texto explicativo opcional y el valor que debe tomar entre //, es decir

```
SCALAR K Coste de capital /0.065/;
```

También puede definirse, sin asignar valor

SCALAR K Coste de capital;

y más adelante, cuando vaya a ser usado, asignarle el valor del siguiente modo

$K=0.65$;

Obsérvese que al final de cada declaración o asignación se termina con ;

PARAMETER es el más versátil de todos los comandos, puesto que puede ser utilizado indistintamente para definir escalares, vectores, matrices, etc. La notación es similar a la de scalar si es para un único valor, si es para un vector se realiza del modo siguiente

PARAMETER PTO(A) Presupuesto para cada año

/A1 -5000

A2 -1000

A3 0

A4 0/;

es decir, con los elementos del vector y sus valores entre /. Más adelante definiremos parameters bidimensionales y tridimensionales.

Los elementos para los cuales no hemos asignado valor, gams los tomará como cero.

TABLE se utiliza fundamentalmente para introducir matrices, su notación es la siguiente

TABLE F(P,A) Flujos netos de caja

	A1	A2	A3	A4	A5	A6	A7	A8	A9
P1	-1100	-200	300	400	560	1200	1100	600	700
P2	-1200	-480	-400	-300	3000	1800	1500	800	300
P3	-2700	-1300	-1000	2800	2500	1800	1000	600	200
P4	-1200	-700	1200	800	400	300	200	100	100
P5	-800	-200	500	400	300	200	200		
P6	-450	-150	300	300	200	150	100		
P7	-1000	-400	800	700	600	400	200		
P8	-500	-50	200	200	100	100	60	20	10
P9			-500	-400	300	200	200	100	100
P10			-800	500	400	300	200	200	
P11				-1000	200	800	400	300	176
P12				-800	200	500	400	300	264
P13					-3000	2000	1000	400	300
P14						-2500	1000	800	400
P15							-3500	1000	3000

El comando table sólo permite la introducción de datos desde el teclado, si queremos calcular unos datos en función de otros necesariamente deberemos recurrir a parameter.

operadores básicos

suma/resta	+	-	x+y	3-x
Producto/cociente	*	/	x*2	x/2
potenciación	**	x**2	ó	power(x,n) power(x,2)
sumatorio		$\sum_i x_i$		Sum[i, x(i)]

También tenemos el operador condicional \$ (si) muy útil a la hora de realizar determinadas operaciones.

En la expresión

scalar a,b;

a = -1;

b=3+a\$(a>0);

b será el resultado de sumarle a 3 el valor de a, si a es mayor que cero, si a es menor o igual que cero b será 3. En este caso b vale 3.

Con estos operadores y alguno más que veremos más adelante podemos plantear una gran variedad de modelos

Por ejemplo si queremos calcular el Van del proyecto P1, la expresión matemática que nos permite calcularlo es:

$$VANP1 = \frac{-1100}{(1+0.065)^0} + \frac{-200}{(1+0.065)^1} + \frac{300}{(1+0.065)^2} + \dots + \frac{700}{(1+0.065)^8}$$

en términos de la nomenclatura utilizada queda

$$VANP1 = \frac{F(P1, A1)}{(1+k)^0} + \frac{F(P1, A2)}{(1+k)^1} + \frac{F(P1, A3)}{(1+k)^2} + \dots + \frac{F(P1, A9)}{(1+k)^8}$$

y en gams se escribiría

$$VANP1 = \frac{F("P1", A1)}{(1+k)^0} + \dots + \frac{F("P1", A9)}{(1+k)^8} = SUM(A, \frac{F("P1", A)}{(1+k)^{ord(A)-1}})$$

y si quisiéramos calcular el van de todos los proyectos, se haría de la misma forma definiendo un parámetro

PARAMETER VAN(P) VAN de cada proyecto;

$$\text{VAN(P)} = \text{SUM}\{A, F(P,A)/[(1+K)**(\text{ORD}(A)-1)]\}$$

3.2.- Bloques obligatorios

a) Bloque variables

Dentro del bloque de variables, lo primero que hay que hacer es declarar que nombres vamos a utilizar para referirnos, dentro del modelo, a cada una de ellas. La declaración debe comenzar obligatoriamente con la palabra variables, especificando el tipo de variables de que se trata, free, positive, negative, integer o binary. La notación a emplear es

VARIABLES

VANT VAN global de la inversión;

Por defecto, si sólo ponemos variables, se entienden libres así VANT es una variable libre

INTEGER VARIABLES IE(PE) proyectos repetitivos no fraccionables ;

Son proyectos no fraccionables pero que pueden realizarse más de una vez, por tanto las variables deben ser enteras

BINARY VARIABLES IB(PB) proyectos ni repetitivos ni fraccionables;

Son proyectos que se han de realizar en su totalidad o no realizarse, por tanto las variables deben ser binarias

Después de declarar el tipo de variables, añadiríamos, si las hay, las cotas sobre ellas. Por ejemplo, si los proyectos repetitivos se han de hacer un mínimo de 2 veces cada uno y un máximo de 4, deberíamos añadir

$$\text{IE.LO(PE)}=2;$$

$$\text{IE.UP(PE)}=4$$

si la cota solo fuera para una variable, por ejemplo el proyecto 10, sería

$$\text{IE.LO("P10")}=2;$$

b) Bloque de ecuaciones

Al igual que en el bloque de variables, lo primero es declarar los nombres que vamos a utilizar para referirnos a cada ecuación, lo cual se hace con el comando equations y a continuación los nombres, finalizando el bloque con ; La notación es la siguiente.

EQUATIONS

VANTOTAL ecuación para el cálculo del van

PRES(A) ecuaciones presupuestarias para cada año;

En el primer caso se trata de una única ecuación, mientras que en el segundo se trata de 10 ecuaciones, una ecuación para cada elemento del conjunto A (Años)

Matemáticamente la ecuación del cálculo del VAN es:

$$VANT = IB(P1) * VAN(P1) + \dots + IB(P8) * VAN(P8) + IE(P9) * VAN(P9) + \dots + IE(P15) * VAN(P15) = \sum_{PB=P1}^{P8} IB(PB) * VAN(PB) + \sum_{PE=P9}^{P15} IE(PE) * VAN(PE)$$

en GAMS sabiendo que los signos de las ecuaciones son ($\leq =L=$), ($= =E=$), ($\geq =G=$) se escribiría como

VANTOTAL.. VANT=E=SUM[PB, IB(PB)*VAN(PB)]+ SUM[PE, IE(PE)*VAN(PE)]

Obsérvese que el nombre de la ecuación va seguido de ..

Respecto a las ecuaciones presupuestarias, matemáticamente su expresión es

$$A1 \quad \sum_{PB} IB_{PB} \cdot F_{PB,A1} + \sum_{PE} IE_{PE} \cdot F_{PE,A1} \geq PTO_{A1}$$

$$A9 \quad \sum_{PB} IB_{PB} \cdot F_{PB,A9} + \sum_{PE} IE_{PE} \cdot F_{PE,A9} \geq PTO_{A9}$$

Ecuaciones que pueden expresarse conjuntamente como

$$\sum_{PB} IB_{PB} \cdot F_{PB,A} + \sum_{PE} IE_{PE} \cdot F_{PE,A} \geq PTO_A \quad \forall A = A1, \dots, A9$$

En GAMS estas 10 ecuaciones se expresan con una única instrucción

PRES(A).. SUM[PB, IB(PB)*F(PB,A)]+ SUM[PE, IE(PE)*F(PE,A)]=G=PTO(A);

La (A) a continuación del nombre de la ecuación hace que se genere una ecuación para cada valor de A.

c) Bloque de modelo

Una vez definidas las variables, ecuaciones, etc. el siguiente paso es definir los modelos que queremos considerar, puesto que en gams es posible definir más de un modelo en cada fichero, lo cual consiste en escribir el comando *model*, seguido del nombre que queremos poner al modelo y entre "/" las ecuaciones que componen dicho modelo separados por comas. Es decir el formato es el siguiente:

MODEL MOD1 /vanttotal, pres/ ;

Cuando el modelo incluye todas las ecuaciones, como es el caso, podemos sustituir la expresión anterior por

MODEL MOD1 /ALL/ ;

d) Bloque de solución

El último paso consiste en decirle al programa que modelos, de los definidos en el bloque anterior, queremos resolver, especificando el tipo de modelo de que se trata, no lineal (nlp), lineal (lp), lineal entero (mip), etc. y si el objetivo es maximizar o minimizar. Para el modelo anterior la instrucción que permite resolverlo es

SOLVE MOD1 USING LP MAXIMIZING VANT;

o en forma abreviada

SOLVE MOD1 US LP MAX VANT;

A continuación tenemos el fichero completo (selecproy.gms) tal como debería ser escrito.

```
OPTION MIP =CPLEX;
OPTION OPTCR=0.001;
option limcol=15;
option limrow=15;
SET P proyectos /P1*P15/
    A años /A1*A9/
    PB(P) proyectos no repetitivos /P1*P8/
    PE(P) proyectos repetitivos /P9*P15/;
SCALAR K Coste de capital /0.065/
PARAMETER PTO(A) Presupuesto para cada año
/A1      -5000
A2      -1000/;
*a partir del año 3 se esperan flujos positivos.
TABLE F(P,A) Flujos netos de caja

```

	A1	A2	A3	A4	A5	A6	A7	A8	A9
P1	-1100	-200	300	400	560	1200	1100	600	700
P2	-1200	-480	-400	-300	3000	1800	1500	800	300
P3	-2700	-1300	-1000	2800	2500	1800	1000	600	200
P4	-1200	-700	1200	800	400	300	200	100	100
P5	-800	-200	500	400	300	200	200		
P6	-450	-150	300	300	200	150	100		
P7	-1000	-400	800	700	600	400	200		
P8	-500	-50	200	200	100	100	60	20	100
P9			-500	-400	300	200	200	100	100
P10			-800	500	400	300	200	200	
P11				-1000	200	800	400	300	1700
P12				-800	200	500	400	300	2800
P13					-3000	2000	1000	400	3000
P14						-2500	1000	800	4000
P15							-3500	1000	3000

```

PARAMETER VAN(P) VAN de cada proyecto;
VAN(P)=SUM{A, F(P,A)/[(1+K)**(ORD(A)-1)]};
DISPLAY VAN;

VARIABLES
VANT van global de la inversión;

BINARY VARIABLES
IB(PB) si se realiza (1) o no (0) el proyecto P;

INTEGER VARIABLES
IE(PE) número de veces que se realiza el proyecto P;

EQUATIONS
VANTOTAL ecuación para el cálculo del van
PRES(A) ecuaciones presupuestarias para cada año;

VANTOTAL.. VANT=E=SUM[PE, IE(PE)*VAN(PE)]+SUM[PB, IB(PB)*VAN(PB)];
PRES(A).. SUM(PB, F(PB,A)*IB(PB))+SUM(PE, F(PE,A)*IE(PE))=G=PTO(A);

MODEL MOD1 /ALL/;
SOLVE MOD1 US MIP MAX VANT;

DISPLAY IE.L, IB.L, VANT.L;

```

4.- Ejecución del programa e interpretación de la solución

Una vez escrito todo el fichero, estamos en condiciones de ejecutar el programa, pulsando “F9” o pinchando con el ratón en el icono con una flecha roja, situado a la derecha del de impresión (ver gráfico siguiente) y obtener la solución.

```

$ontext
Ejemplo de selección de proyectos de inversión repetitivos
y no repetitivos con limitaciones presupuestarias
$offtext

OPTION MIP =CPLEX;
OPTION OPTCR=0.001;
option limcol=15;
option limrow=15;
SET P proyectos /P1*P15/
    A años /A1*A9/
    PB(P) proyectos no repetitivos /P1*P8/
    PE(P) proyectos repetitivos /P9*P15/;
SCALAR K Coste de capital /0.065/
PARAMETER PTO(A) Presupuesto para cada año
/A1 -5000

```

```

IDE gamside: C:\CGyF\project.gpr - [C:\CGyF\selecproy.gms]
IDE File Edit Search Windows Utilities Model Libraries Help
vd13
selecproy.gms
$ontext
Ejemplo de selección de proyectos de inversión repetitivos
y no repetitivos con limitaciones presupuestarias
$offtext

OPTION MIP =CPLEX;
OPTION OPTCR=0.001;
option limcol=15;
option limrow=15;

SET P proyectos /P1*P15/
      A años /2009*2017/
      PB(P) proyectos no repetitivos /P1*P8/
      PE(P) proyectos repetitivos /P9*P15/;

SCALAR K Coste de capital /0.065/

PARAMETER PTO(A) Presupuesto para cada año
/2009 -5000
100:1 Insert

```

al ejecutar aparecerá una ventana como la siguiente

```

IDE gamside: C:\CGyF\project.gpr
File Edit Search Windows Utilities Model Libraries Help
No active process
selecproy
Proven optimal solution.

MIP Solution:      6321.189790      (17 iterations, 0 nodes)
Final Solve:      6321.189790      (0 iterations)


Best possible:    6321.189790
Absolute gap:     0.000000
Relative gap:     0.000000

Close Open Log  Summary only  Update
Column
Model Statistics SOLVE TIR L 11 A años /A1*A9/
Solution Report SOLVE TIR L 12 PB(P) proyectos no repetitivos /P1*P8/
SolEQU 13 PE(P) proyectos repetitivos /P9*P15/;
SolVAR 14 SCALAR K Coste de capital /0.065/
Execution
1:1 Double-Click to Open File

```

si cerramos esta ventana pinchando sobre el icono *close*, volveremos a una pantalla igual que la del fichero gms pero con dos pestañas la del fichero selecproy.gms y otra

pestaña, que será la ventana activa, con el fichero selecproy.lst que es el fichero donde se encuentra la solución, tal como se aprecia en el gráfico siguiente



El fichero LST nos indicará, caso de haber cometido errores, cuáles son dichos errores y donde los hemos cometido. Si no hay errores, contiene además de la solución del modelo, abundante información acerca de las variables y ecuaciones empleadas en el modelo (puede ser de ayuda para comprobar si hemos generado correctamente las ecuaciones). El tiempo de CPU necesario para compilar, generar y ejecutar el modelo así como la memoria utilizada.

De todo el fichero selecproy.lst a continuación reproducimos las partes donde se encuentra la solución.

	LOWER	LEVEL	UPPER	MARGINAL
---- EQU VANTOTAL	.	.	.	1.000
VANTOTAL ecuación para el cálculo del van				
---- EQU PRES	ecuaciones presupuestarias para cada año			
	LOWER	LEVEL	UPPER	MARGINAL
A1	-5000.000	-3850.000	+INF	.
A2	-1000.000	-1000.000	+INF	.
A3	.	500.000	+INF	.
A4	.	.	+INF	.
A5	.	160.000	+INF	.
A6	.	7050.000	+INF	.
A7	.	760.000	+INF	.
A8	.	3320.000	+INF	.
A9	.	4538.000	+INF	.

	LOWER	LEVEL	UPPER	MARGINAL
---- VAR VANT	-INF	6321.190	+INF	.
VANT van global de la inversión				
---- VAR IB	si se realiza (1) o no (0) el proyecto P			
	LOWER	LEVEL	UPPER	MARGINAL
P1	.	1.000	1.000	2181.935
P2	.	.	1.000	3118.108
P3	.	.	1.000	1965.038
P4	.	.	1.000	654.727
P5	.	1.000	1.000	300.416
P6	.	1.000	1.000	255.488
P7	.	1.000	1.000	804.648
P8	.	1.000	1.000	5.706
---- VAR IE	número de veces que se realiza el proyecto P			
	LOWER	LEVEL	UPPER	MARGINAL
P9	.	.	+INF	-130.956
P10	.	2.000	+INF	504.259
P11	.	3.000	+INF	485.050
P12	.	.	+INF	484.828
P13	.	1.000	+INF	251.798
P14	.	.	+INF	-382.871
P15	.	1.000	+INF	57.530
**** REPORT SUMMARY :	0 NONOPT			
	0 INFEASIBLE			
	0 UNBOUNDED			

Vamos a comenzar leyendo este fichero de abajo hacia arriba,

En primer lugar report summary

```
**** REPORT SUMMARY :      0      NONOPT
                          0 INFEASIBLE
                          0 UNBOUNDED
```

En este apartado deberá aparecer todo cero, de lo contrario significa que el problema no tiene óptimo, que es no factible o no está acotado, por lo que deberíamos revisar el planteamiento del problema, por si hemos cometido algún error en la modelización. Por supuesto los valores que aparecen en estos casos no tienen ningún sentido.

A continuación vamos a comentar el bloque variables

	LOWER	LEVEL	UPPER	MARGINAL
P1	.	1.000	1.000	2181.935
P2	.	.	1.000	3118.108
P3	.	.	1.000	1965.038
P4	.	.	1.000	654.727
P5	.	1.000	1.000	300.416
P6	.	1.000	1.000	255.488
P7	.	1.000	1.000	804.648
P8	.	1.000	1.000	5.706

En primer lugar señalar que GAMS utiliza el punto como separador de decimales con tres decimales y la coma como separador unidades de millar por lo que 1.000 =1 (no confundir con mil). Por otra parte sustituye el valor cero 0.000 por un punto (columna lower variables)

Cada variable o ecuación tiene, como podemos apreciar, cuatro columnas, **lower** (cota inferior), **level** valor en la solución, **upper** (cota superior) y **marginal**.

Como era de esperar, dado que hemos declarado las variables como binarias su cota inferior es cero, de ahí que en la columna lower todas las variables aparezcan con un punto y como cota superior 1 (UPPER)

La columna level refleja el valor de cada variable y por tanto indica que la empresa debería acometer los proyectos P1,P5,P6,P7 y P8

Respecto a la columna marginal, para problemas lineales continuos, *si es distinta de cero*, representa el coste de oportunidad de cada variable, es decir como se comportará la función objetivo VAN respecto a cambios en el valor de la variable correspondiente de acuerdo con la expresión

$$\Delta VANT = \Delta IE_{PE} * Mg_{PE}$$

Con las restantes variables al ser enteras su cota inferior (level) es cero y su cota superior (upper) es 100, en realidad debería ser infinito, dado que no hemos puesto ninguna cota superior, pero GAMS se autoimpone esta cota para facilitar cálculos, caso de poder tomar valores superiores a 100, deberíamos establecer cotas lo suficientemente grandes.

VAR	IE	cantidad invertida en el proyecto	P	
	LOWER	LEVEL	UPPER	MARGINAL
P9	.	.	100.000	-130.956
P10	.	2.000	100.000	504.259
P11	.	3.000	100.000	485.050
P12	.	.	100.000	484.828
P13	.	1.000	100.000	251.798
P14	.	.	100.000	-382.871
P15	.	1.000	100.000	57.530

El Van total obtenido es de 6321,19 millones de €

VAR	VANT	LOWER	LEVEL	UPPER	MARGINAL
----	VAR VANT	-INF	6321.190	+INF	.

Respecto a las ecuaciones presupuestarias en la columna lower aparece el presupuesto para cada año, mientras que el valor de la columna level representa los flujos reales obtenidos con la inversión

---- EQU PRES ecuaciones presupuestarias para cada año				
	LOWER	LEVEL	UPPER	MARGINAL
A1	-5000.000	-3850.000	+INF	.
A2	-1000.000	-1000.000	+INF	.
A3	.	500.000	+INF	.
A4	.	.	+INF	.
A5	.	160.000	+INF	.
A6	.	7050.000	+INF	.
A7	.	760.000	+INF	.
A8	.	3320.000	+INF	.
A9	.	4538.000	+INF	.

Así podemos apreciar que en el momento inicial (A1) se ha precisado de un desembolso de 3850 millones de euros mientras que la disponibilidad era de 5000, en el segundo (A2) sólo se ha utilizado 1000 millones que era la totalidad del presupuesto disponible para ese año, en el tercero (A3) se obtienen unos flujos netos de caja de 500 millones y así sucesivamente.

Selección de proyectos

S&R Asociados son una empresa de consultoría, dedicada al análisis de alternativas de Inversión para diferentes clientes. Harrods S.A. es una empresa que ha presentado cinco alternativas de inversión para el estudio por parte de la consultoría.

El trabajo de la consultoría se ha limitado a evaluar los flujos netos de caja que va a producir cada inversión a lo largo del horizonte de planificación y son los siguientes (en millones de euros)

	1	2	3	4	5	6	7	8
X2	-50	-35	-45	60	50	50	40	30
4E	-20	-35	-25	40	40	30	20	0
ALFA	-10	-25	-20	25	25	25	0	0
BETA	-35	-45	-30	50	40	30	30	30
R3	-40	-10	-25	20	20	20	20	20

El presupuesto que Harrods puede destinar a la inversión es de 150 millones de euros el primer año, 125 millones el segundo y 125 millones el tercer año, a partir del cuarto año no hay presupuesto y el conjunto de las inversiones deberían empezar a ser productivas.

- 1) Determinar las inversiones a realizar.
- 2) Determinar las inversiones a realizar si la inversión x2 no es viable si no se hace junto con la Alfa, aunque ésta última puede hacerse individualmente.
- 3) Determinar las inversiones a realizar sabiendo que la inversión x2 y Beta son incompatibles